

Ejercicio 3

- Análisis

- Entrada

cod

medEsp

medAct

- Relaciones

$\text{error} = \text{abs}(\text{medAct} - \text{medEsp})$

$\text{totFallas} = \text{totFallas} + \text{fallas}$

- Salida

"Código del lote (0 para terminar):"

"Medida esperada para el lote"

"Ingrese medida del tornillo"

"Resumen Lote"

"Mayor error:"

"Porcentaje de fallas:"

"RESULTADOS FINALES"

"Total lotes procesados:"


"% general de fallas:"

"Mejor lote:"

"Peor lote:"

"No se procesaron lotes."

- Estrategia



- Ambiente

Variable	Tipo	Descripción
cod	Entero	Código del lote
loteMenor	Entero	Nº de lote con menor fallas
loteMayor	Entero	Nº de lote con mayor fallas
medEsp	Real	Medida esperada del lote
medAct	Real	Medida del tornillo
maxError	Real	Error máximo
error	Real	Diferencia entre medida esperada y medida real
lotes	Entero	Total de lotes

fallas	Entero	Numero de tornillos con fallas
totFallas	Entero	Numero total de fallas
minFallas	Entero	Falla menor
maxFallas	Entero	Falla mayor

- Pseudocodigo

```

1  Proceso ControlCalidadTornillos
2  Definir cod, loteMenor, loteMayor Como Entero;
3  Definir medEsp, medAct, maxError, error Como Real;
4  Definir lotes, fallas, totFallas, minFallas, maxFallas Como Entero;
5  Definir i Como Entero;
6  lotes ← 0;
7  totFallas ← 0;
8  minFallas ← 11;
9  maxFallas ← -1;
10 Repetir
11     Escribir 'Código del lote (0 para terminar):';
12     Leer cod;
13     Si cod<>0 Entonces
14         lotes ← lotes+1;
15         fallas ← 0;
16         maxError ← 0;
17         Escribir 'Medida esperada para el lote ', cod, ':';
18         Leer medEsp;
19         Para i←1 Hasta 10 Hacer
20             Escribir 'Ingrese medida del tornillo ', i, ':';
21             Leer medAct;
22             error ← abs(medAct-medEsp);
23             Si error>maxError Entonces
24                 maxError ← error;
25             FinSi
26             Si error>0 Entonces
27                 fallas ← fallas+1;
28             FinSi
29         FinPara
30         totFallas ← totFallas+fallas;
31         Si fallas<minFallas Entonces
32             minFallas ← fallas;
33             loteMenor ← cod;
34         FinSi
35         Si fallas>maxFallas Entonces
36             maxFallas ← fallas;
37             loteMayor ← cod;
38         FinSi
39         Escribir 'Resumen Lote ', cod, ':';
40         Escribir 'Mayor error: ', maxError;
41         Escribir 'Porcentaje de fallas: ', (fallas*10), '%';
42     FinSi
43 Hasta Que cod=0
44 Si lotes>0 Entonces
45     Escribir 'RESULTADOS FINALES';
46     Escribir 'Total lotes procesados: ', lotes;
47     Escribir '% general de fallas: ', (totFallas*10)/lotes, '%';
48     Escribir 'Mejor lote: ', loteMenor, ' (', minFallas, ' fallas)';
49     Escribir 'Peor lote: ', loteMayor, ' (', maxFallas, ' fallas)';
50 SiNo
51     Escribir 'No se procesaron lotes.';
52 FinSi
53 FinProceso

```

- Diagrama de flujo

